$\mathcal{KL}$ as a Knowledge Base Logic in Haskell

Natasha De Kriek, Milan Hartwig, Victor Joss, Paul Weston, Louise Wilk

Tuesday 18th March, 2025

Abstract

In this project, we aim to implement the first-order epistemic logic $\mathcal{KL}$ as introduced by Levesque (1981) and refined by Levesque and Lakemeyer (2001). The semantics for this logic evaluates formulae on world states and epistemic states where world states are sets of formulae that are true at the world and epistemic states are sets of world states that are epistemically accessible. Levesque and Lakemeyer use the language $\mathcal{KL}$ as "a way of communicating with a knowledge base" (ibid. p. 79). For this, they define an ASK- and a TELL-operation on a knowledge base. In our project, we implement a $\mathcal{KL}$-model, the ASK- and TELL- operations, a tableau-based satisfiablity and validity checking for $\mathcal{KL}$, as well as compare $\mathcal{KL}$-models to epistemic Kripke models and implement a translation function between them.

Contents

1 $\mathcal{KL}$: Syntax and Semantics

1.1 Syntax of $\mathcal{KL}$

The syntax of the language $\mathcal{KL}$ is described in Levesque and Lakemeyer (2001) and inspired by Levesque's work (Levesque 1981). The SyntaxKL module establishes the foundation for $\mathcal{KL}$'s syntax, defining the alphabet and grammar used in subsequent semantic evaluation.

```
{-# LANGUAGE InstanceSigs #-}


module SyntaxKL where

import Test.QuickCheck
```

Symbols of $\mathcal{KL}$

The expressions of $\mathcal{KL}$ are constituted by sequences of symbols drawn from the following two sets (cf. Levesque 1981): Firstly, the *logical symbols*, which consist of the logical connectives and quantifiers $\exists, \vee, \neg$, as well as punctuation and parentheses. Furthermore, it compromises a countably infinite supply of first-order variables denoted by the set $\{x, y, z, \ldots\}$, a countably infinite supply of standard names, represented by the set $\{\#1, \#2, \ldots\}$, and the equality symbol $=$. The *non-logical symbols* comprise predicate symbols of any arity $\{P, Q, R, \ldots\}$, which are intended to represent domain-specific properties and relations, and function symbols of any arity, which are used to denote mappings from individuals to individuals (Levesque and Lakemeyer 2001, p.22).

n this implementation, standard names are represented as strings (e.g., "n1", "n2") via the StdName type, and variables are similarly encoded as strings (e.g., "x", "y") with the Variable type, ensuring that we have a distinct yet infinite supplies of each.

```
arbitraryUpperLetter :: Gen String
arbitraryUpperLetter = (:[]) <$> elements ['A'..'Z']

arbitraryLowerLetter :: Gen String
arbitraryLowerLetter = (:[]) <$> elements ['a'..'z']

-- Represents a standard name (e.g., "n1") from the infinite domain N
newtype StdName = StdName String deriving (Eq, Ord, Show)
instance Arbitrary StdName where
  arbitrary:: Gen StdName
  arbitrary = StdName . ("n" ++) . show <$> elements [1 .. 20::Int]

-- Represents a first-order variable (e.g., "x")
newtype Variable = Var String deriving (Eq, Ord, Show)
instance Arbitrary Variable where
  arbitrary:: Gen Variable
  arbitrary = Var . show <$> elements [1 .. 20::Int]
```

Terms and Atoms

Terms in $\mathcal{KL}$ are the building blocks of expressions, consisting of variables, standard names, or function applications. Atomic propositions (atoms) are formed by applying predicate symbols to lists of terms. To distinguish primitive terms (those that contain no variable and only a single function symbol) and primitive atoms (those atoms that contain no variables and only standard names as terms) for semantic evaluation, we also define PrimitiveTerm and PrimitiveAtom.

```
-- Defines terms: variables, standard names, or function applications
data Term = VarTerm Variable    -- A variable (e.g., "x")
          | StdNameTerm StdName    -- A standard name (e.g., "n1")
          | FuncAppTerm String [Term]    -- Function application (e.g., "Teacher" ("x"))
          deriving (Eq, Ord, Show)
```

```
instance Arbitrary Term where
  arbitrary :: Gen Term
  arbitrary = sized $ \n -> genTerm (min n 5) where
    genTerm 0 = oneof [VarTerm <$> arbitrary,
                       StdNameTerm <$> arbitrary]
    genTerm n = oneof [VarTerm <$> arbitrary,
                       StdNameTerm <$> arbitrary,
                       FuncAppTerm <$> arbitraryLowerLetter
                                   <*> resize (n `div` 2) (listOf1 (genTerm (n `div` 2)))]

-- Terms with no variables and only a single function symbol
data PrimitiveTerm = PStdNameTerm StdName   -- e.g., "n1"
                   | PFuncAppTerm String [StdName]
  deriving (Eq, Ord, Show)

-- Define Atoms as predicates applied to terms
data Atom = Pred String [Term]  --e.g. "Teach" ("n1", "n2")
  deriving (Eq, Ord, Show)

instance Arbitrary Atom where
  arbitrary :: Gen Atom
  arbitrary = sized $ \n -> genAtom (min n 5) where
      genAtom :: Int -> Gen Atom
      genAtom 0 = Pred <$> arbitraryLowerLetter <*> pure []
      genAtom n = Pred <$> arbitraryLowerLetter <*> vectorOf n arbitrary

-- Atoms with only standard names as terms
data PrimitiveAtom = PPred String [StdName]
  deriving (Eq, Ord, Show)
```

Formulas

$\mathcal{KL}$-formulas are constructed recursively from atoms, equality, and logical operators. The Formula type includes atomic formulas, equality between terms, negation, disjunction, existential quantification, and the knowledge operator K. Additional connectives like universal quantification ($\forall$), implication ($\rightarrow$), and biconditional ($\leftrightarrow$) are defined as derived forms for convenience.

```
--Defines KL-formulas with logical and epistemic constructs
data Formula = Atom Atom                -- Predicate (e.g. Teach(x, "n1"))
             | Equal Term Term          -- Equality (e.g., x = "n1")
             | Not Formula              -- Negation
             | Or Formula Formula       -- Disjunction
             | Exists Variable Formula -- Existential (e.g., exists x (Teach x "sue"))
             | K Formula                -- Knowledge Operator (e.g., K (Teach "ted" "sue"))
             deriving (Eq, Ord, Show)

instance Arbitrary Formula where
  arbitrary :: Gen Formula
  arbitrary = sized $ \n -> genFormula (min n 5)   where
    genFormula 0 = oneof [Atom <$> arbitrary,
                          Equal <$> arbitrary <*> arbitrary]
    genFormula n = oneof [Not <$> genFormula (n `div` 2),
                          Or <$> genFormula (n `div` 2) <*> genFormula (n `div` 2),
                          Exists <$> arbitrary <*> genFormula (n `div` 2),
                          K <$> genFormula (n `div` 2)]

-- Universal quantifier as derived form
klforall :: Variable -> Formula -> Formula
klforall x f = Not (Exists x (Not f))

-- Implication as derived form
implies :: Formula -> Formula -> Formula
implies f1 = Or (Not f1)

-- Biconditional as derived form
iff :: Formula -> Formula -> Formula
iff f1 f2 = Or (Not (Or f1 f2)) (Or (Not f1) f2)
```

We can now use this implementation of $\mathcal{KL}$'s syntax to implement the semantics.

1.2 Semantics of $\mathcal{KL}$

$\mathcal{KL}$ is an epistemic extension of first-order logic designed to model knowledge and uncertainty, as detailed in Levesque and Lakemeyer (2001). It introduces a knowledge operator K and uses an infinite domain $\mathcal{N}$ of standard names to denote individuals. Formulas are evaluated in world states: consistent valuations of atoms and terms, while epistemic states capture multiple possible worlds, reflecting epistemic possibilities.
The semantics are implemented in the SemanticsKL module, which imports syntactic definitions from SyntaxKL and uses Haskell's Data.Map and Data.Set for efficient and consistent mappings.

```
{-# LANGUAGE InstanceSigs #-}

module SemanticsKL where

import SyntaxKL
import Data.Map (Map)
import qualified Data.Map as Map
import Data.Set (Set)
import qualified Data.Set as Set

import Test.QuickCheck
```

Worlds and Epistemic States

A WorldState represents a single possible world in $\mathcal{KL}$, mapping truth values to primitive atoms and standard names to primitive terms. An EpistemicState, defined as a set of WorldStates, models the set of worlds an agent considers possible, enabling the evaluation of the K operator.

```
-- A single world state with valuations for atoms and terms
data WorldState = WorldState
 { atomValues :: Map Atom Bool,     -- Maps (primitive) atoms to truth values
   termValues :: Map Term StdName   --Maps (primitive) terms to standard names
  }  deriving (Eq, Ord, Show)

instance Arbitrary WorldState where
  arbitrary :: Gen WorldState
  arbitrary = WorldState <$> arbitrary <*> arbitrary

-- A set of possible world states, modeling epistemic possibilities
type EpistemicState = Set WorldState
```

Constructing World States

We can construct world states by using mkWorldState, which builds a WorldState from lists of primitive atoms and terms. While a WorldState is defined in terms of Atom and Term, we use mkWorldState to make sure that we can only have primitive atoms and primitive terms in the mapping. To be able to use primitive terms and atoms in other functions just as we would use atoms and terms (since primitive atoms and primitive terms are atoms and terms as well), we convert the constructors to those of regular terms and atoms. We then use the function checkDups to ensure that there are no contradictions in the world state (e.g., P(n1) mapped to both True and False), thus reinforcing the single-valuation principle (Levesque and Lakemeyer 2001, p. 24). mkWorldState then constructs maps for efficient lookup.

```
-- Constructs a WorldState from primitive atoms and primitive terms
mkWorldState :: [(PrimitiveAtom, Bool)] -> [(PrimitiveTerm, StdName)] -> WorldState
mkWorldState atoms terms =
  let convertAtom (PPred p ns, b) = (Pred p (map StdNameTerm ns), b)  -- Convert primitive
      atom to Atom
```

```
      convertTerm (PStdNameTerm n, v) = (StdNameTerm n, v)  -- Convert primitive term to
          Term
      convertTerm (PFuncAppTerm f ns, v) = (FuncAppTerm f (map StdNameTerm ns), v)
      atomList = map convertAtom atoms
      termList = map convertTerm terms
  in WorldState (Map.fromList (checkDups atomList)) (Map.fromList (checkDups termList))

-- Checks for contradictory mappings in a key-value list
checkDups :: (Eq k, Show k, Eq v, Show v) => [(k, v)] -> [(k, v)]
checkDups [] = []  -- Empty list is consistent
checkDups ((k, v) : rest) =  -- Recursively checks each key k against the rest of the list.
  case lookup k rest of
    Just v' | v /= v' -> error $ "Contradictory mapping for " ++ show k ++ ": " ++ show v
        ++ " vs " ++ show v' -- If k appears with a different value v', throws an error.
    _ -> (k, v) : checkDups rest  -- Keep pair if no contradiction
```

Since we have decided to change the constructors of data of type PrimitiveAtom or PrimitiveTerm to those of Atom and Term, we have implemented two helper-functions to check if a Term or an Atom is primitive. This way, we can, if needed, check whether a given term or atom is primitive and then change the constructors appropriately.

```
-- Checks if a term is primitive (contains only standard names)
isPrimitiveTerm :: Term -> Bool
isPrimitiveTerm (StdNameTerm _) = True
isPrimitiveTerm (FuncAppTerm _ args) = all isStdName args
  where isStdName (StdNameTerm _) = True
        isStdName _ = False
isPrimitiveTerm _ = False

-- Checks if an atom is primitive
isPrimitiveAtom :: Atom -> Bool
isPrimitiveAtom (Pred _ args) = all isStdName args
  where isStdName (StdNameTerm _) = True
        isStdName _ = False
```

Term Evaluation

To evaluate a ground term in a world state, we define a function evalTerm that takes a WorldState and a Term and returns a StdName. The idea is to map syntactic terms to their semantic values (standard names) in a given world state. The function uses pattern matching to handle the three possible forms of Term:

1. VarTerm _
   If the term is a variable (e.g., x), it throws an error. This enforces a precondition that evalTerm only works on ground terms (terms with no free variables). In $\mathcal{KL}$, variables must be substituted with standard names before evaluation, aligning with the semantics where only ground terms have denotations (Levesque and Lakemeyer 2001, p. 24). This is a runtime check to catch ungrounded inputs.

2. StdNameTerm n
   If the term is a standard name wrapped in StdNameTerm (e.g., StdNameTerm (StdName "n1")), it simply returns the underlying StdName (e.g., StdName "n1"). Standard names in $\mathcal{KL}$ are constants that denote themselves (ibid., p.22). For example, if n=StdName "n1", it represents the individual n1, and its value in any world is n1. In this case, no lookup or computation is needed.

3. FuncAppTerm f args
   If the term is a function application (e.g., f(n1,n2)), evalTerm evaluates the argument, by recursively computing the StdName values of each argument in args using evalTerm

w. Next, the ground term is constructed: It Builds a new FuncAppTerm term where all arguments are standard names (wrapped in StdNameTerm), ensuring it's fully ground. We then look up the value by querying the termValues map in the world state w for the denotation of this ground term, erroring on undefined terms.

```
-- Evaluates a ground term to its standard name in a WorldState
evalTerm :: WorldState -> Term -> StdName
evalTerm w t = case t of
  VarTerm _ -> error "evalTerm: Variables must be substituted"  -- Variables are not ground
  StdNameTerm n -> n   -- Standard names denote themselves
  FuncAppTerm f args ->
    let argValues = map (evalTerm w) args   -- Recursively evaluate arguments
        groundTerm = FuncAppTerm f (map StdNameTerm argValues)   -- Construct ground term
    in case Map.lookup groundTerm (termValues w) of
        Just n -> n   -- Found in termValues
        Nothing -> error $ "evalTerm: Undefined ground term " ++ show groundTerm   -- Error
             if undefined
```

Groundness and Substitution

To support formula evaluation, isGround and isGroundFormula check for the absence of variables, while substTerm and subst perform substitution of variables with standard names, respecting quantifier scope to avoid capture. We need these functions to be able to define a function that checks whether a formula is satisfiable in a worldstate and epistemic state.

```
-- Check if a term is ground (contains no variables).
isGround :: Term -> Bool
isGround t = case t of
  VarTerm _ -> False
  StdNameTerm _ -> True
  FuncAppTerm _ args -> all isGround args

-- Check if a formula is ground.
isGroundFormula :: Formula -> Bool
isGroundFormula f = case f of
  Atom (Pred _ terms) -> all isGround terms
  Equal t1 t2 -> isGround t1 && isGround t2
  Not f' -> isGroundFormula f'
  Or f1 f2 -> isGroundFormula f1 && isGroundFormula f2
  Exists _ _ -> False   -- always contains a variable
  K f' -> isGroundFormula f'

-- Substitute a variable with a standard name in a term.
substTerm :: Variable -> StdName -> Term -> Term
substTerm x n t = case t of
  VarTerm v | v == x -> StdNameTerm n -- Replace variable with name
  VarTerm _ -> t
  StdNameTerm _ -> t
  FuncAppTerm f args -> FuncAppTerm f (map (substTerm x n) args)

-- Substitute a variable with a standard name in a formula.
subst :: Variable -> StdName -> Formula -> Formula
subst x n formula = case formula of
  Atom (Pred p terms) -> Atom (Pred p (map (substTerm x n) terms))
  Equal t1 t2 -> Equal (substTerm x n t1) (substTerm x n t2)
  Not f -> Not (subst x n f)
  Or f1 f2 -> Or (subst x n f1) (subst x n f2)
  Exists y f | y == x -> formula -- Avoid capture
             | otherwise -> Exists y (subst x n f)
  K f -> K (subst x n f)
```

Model and Satisfiability

Since we want to check for satisfiability in a model, we want to make the model explicit:

```
-- Represents a model with an actual world, epistemic state, and domain
data Model = Model
```

```
  { actualWorld :: WorldState       -- The actual world state
  , epistemicState :: EpistemicState -- Set of possible world states
  , domain :: Set StdName           -- Domain of standard names
  } deriving (Show)

instance Arbitrary Model where
  arbitrary:: Gen Model
  arbitrary = Model <$> arbitrary <*> arbitrary <*> arbitrary
```

A Model encapsulates an actual world, an epistemic state, and a domain, enabling the evaluation of formulas with the Koperator. satisfiesModel implements $\mathcal{KL}$'s satisfaction relation, checking truth across worlds.

```
-- Checks if a formula is satisfied in a model
satisfiesModel :: Model -> Formula -> Bool
satisfiesModel m = satisfies (epistemicState m) (actualWorld m)
  where
    satisfies e w formula = case formula of
      Atom (Pred p terms) ->
        if all isGround terms
          then Map.findWithDefault False (Pred p terms) (atomValues w)   -- Default False
              for undefined atoms
          else error "Non-ground atom in satisfies!"
      Equal t1 t2 ->
        if isGround t1 && isGround t2   -- Equality of denotations
          then evalTerm w t1 == evalTerm w t2
          else error "Non-ground equality in satisfies!"
      Not f ->
        not (satisfies e w f)
      Or f1 f2 ->
        satisfies e w f1 || satisfies e w f2
      Exists x f ->
      -- \(e, w \models \exists x. \alpha\) iff for some name \(n\), \(e, w \models \
          alpha_n^x\)
        any (\n -> satisfies e w (subst x n f)) (Set.toList $ domain m)
      -- \(e, w \models K \alpha\) iff for every \(w' \in e\), \(e, w' \models \alpha\)
      K f ->
        all (\w' -> satisfies e w' f) e
```

Grounding and Model Checking

Building on this we can implement a function checkModel that checks whether a formula holds in a given model. checkModel ensures a formula holds by grounding it with all possible substitutions of free variables, using groundFormula and freeVars to identify and replace free variables systematically.

```
-- Checks if a formula holds in a model by grounding it
checkModel :: Model -> Formula -> Bool
checkModel m phi = all (satisfiesModel m) (groundFormula phi (domain m))
```

Note that we use the function groundFormula here. Since we have implemented satisfiesModel such that it assumes ground formulas or errors out, we decided to handle free variables by grounding formulas, given a set of free standard names to substitute. Alternatives, would be to error or always substitute the same standard name. The the implementation that we have chosen is more flexible and allows for more varied usage, however it is computationally expensive (We would appreciate it if you have suggestions to improve this). We implement groundFormula as follows:

```
-- Generates all ground instances of a formula
groundFormula :: Formula -> Set StdName -> [Formula]
groundFormula f dom = groundFormula' f >>= groundExists dom
  where
    -- Ground free variables at the current level
    groundFormula' formula = do
      let fvs = Set.toList (freeVars formula)
```

```
      subs <- mapM (\_ -> Set.toList dom) fvs
      return $ foldl (\acc (v, n) -> subst v n acc) formula (zip fvs subs)

    -- Recursively eliminate Exists in a formula
    groundExists domainEx formula = case formula of
      Exists x f' -> map (\n -> subst x n f') (Set.toList domainEx) >>= groundExists
          domainEx
      Atom a -> [Atom a]
      Equal t1 t2 -> [Equal t1 t2]
      Not f' -> map Not (groundExists domainEx f')
      Or f1 f2 -> do
        g1 <- groundExists domainEx f1
        g2 <- groundExists domainEx f2
        return $ Or g1 g2
      K f' -> map K (groundExists domainEx f')
```

This function takes a formula and a domain of standard names and returns a list of all possible ground instances of the formula by substituting its free variables with elements from the domain. We use a function variables that identifies all the variables in a formula that need grounding or substitution. If the Boolean 'includeBound' is 'True', variables returns all variables (free and bound) in the formula. If 'includeBound' is 'False', it returns only free variables, excluding those bound by quantifiers. This way, we can use the function to support both 'freeVars' (free variables only) and 'allVariables' (all variables).

```
-- Collects variables in a formula, with a flag to include bound variables
variables :: Bool -> Formula -> Set Variable
variables includeBound = vars
  where
    -- Helper function to recursively compute variables in a formula
    vars formula = case formula of
      -- Union of variables from all terms in the predicate
      Atom (Pred _ terms) -> Set.unions (map varsTerm terms)
      -- Union of variables from both terms in equality
      Equal t1 t2 -> varsTerm t1 `Set.union` varsTerm t2
      Not f' -> vars f'
      Or f1 f2 -> vars f1 `Set.union` vars f2
      Exists x f' -> if includeBound
                     then Set.insert x (vars f') -- Include bound variable x if
                         includeBound is True
                     else Set.delete x (vars f') -- Exclude bound variable x if
                         includeBound is False
      K f' -> vars f'  -- Variables in the subformula under K (no binding)

    varsTerm term = case term of
      VarTerm v -> Set.singleton v  -- A variable term contributes itself to the set
      StdNameTerm _ -> Set.empty  -- A standard name has no variables
      FuncAppTerm _ args -> Set.unions (map varsTerm args)  -- Union of variables from all
          function arguments

-- Collects free variables in a formula
freeVars :: Formula -> Set Variable
freeVars = variables False

-- Collects all variables (free and bound) in a formula
allVariables :: Formula -> Set Variable
allVariables = variables True
```

2 Ask and Tell Operators

To use $\mathcal{KL}$ to interact with a knowledge base, Levesque and Lakemeyer (2001) defines two operators on epistemic states: ASK and $TELL$. Informally, ASK is used to determine if a sentence is known to a knowledge base whereas $TELL$ is used to add a sentence to the knowledge

base. Since epistemic states are sets of possible worlds, the more sentences that are known reduces the set of possible worlds. For this purpose, an $INITIAL$ epistemic state is also defined to contain all possible worlds given a finite set of atoms and terms.

```
module AskTell (ask,askModel, tell, tellModel, initial) where
import qualified Data.Set as Set
import SyntaxKL
import SemanticsKL
```

The ASK operator determines whether a formula is known to a knowledge base. Formally, given an epistemic state e and any sentence α of $\mathcal{KL}$,

$$ASK[e, \alpha] = \begin{cases} True & \text{if } e \models K\alpha \\ False & \text{otherwise} \end{cases}$$

When implementing ASK in Haskell, we must take into account that a domain is implied by "$\models$" so that we can evaluate sentences with quantifiers. As such, we will take a domain as our first argument.

```
-- ASK (Definition 5.2.1)
ask :: Set.Set StdName -> EpistemicState -> Formula -> Bool
ask d e alpha | Set.null e = False
              | otherwise = satisfiesModel newModel (K alpha) where
              newModel = Model {actualWorld = (Set.findMin e), epistemicState = e, domain =
                   d}
```

We can simplify this into an askModel function that takes only a model and sentence as input.

```
askModel :: Model -> Formula -> Bool
askModel m alpha |  Set.null (epistemicState m) = False
            |  otherwise = satisfiesModel m (K alpha)
```

The second operation, $TELL$, asserts that a sentence is true and in doing so reduces which worlds are possible. In practice, $TELL[\varphi, e]$ filters the epistemic state e to worlds where the sentence φ holds. That is,

$$TELL[\varphi, e] = e \cap \{w \mid w \models \varphi\}$$

Again, we run into the issue that "$\models$" requires a domain, and so a domain must be specified to evaluate sentences with quantifiers.

```
-- TELL operation (Definition 5.5.1)
tell :: Set.Set StdName -> EpistemicState -> Formula -> EpistemicState
tell d e alpha = Set.filter filterfunc e where
    filterfunc = (\w -> satisfiesModel (Model {actualWorld = w, epistemicState = e, domain
        = d}) alpha)
```

We can again simplify to a function tellModel, that takes as input a model and formula and produces a model with a modified epistemic state.

```
tellModel :: Model -> Formula -> Model
tellModel m alpha = Model {actualWorld = actualWorld m, epistemicState = Set.filter
    filterfunc  (epistemicState m), domain = domain m} where
    filterfunc = (\w -> satisfiesModel (Model {actualWorld = w, epistemicState =
        epistemicState m, domain = domain m}) alpha)
```

In addition to ASK and $TELL$, it is valuable to define an inital epistemic state. $INITIAL$ is the epistemic state before any $TELL$ operations. This state contains all possible world states as there is nothing known that eliminates any possible world.

```
-- INITIAL operation (Section 5.3)
-- Generate all possible world states for a finite set of atoms and terms
allWorldStates :: [PrimitiveAtom] -> [PrimitiveTerm] -> [StdName] -> [WorldState]
allWorldStates atoms terms dom = do
   atomVals <- mapM (\_ -> [True, False]) atoms
   termVals <- mapM (\_ -> dom) terms
   return $ mkWorldState (zip atoms atomVals) (zip terms termVals)

initial :: [PrimitiveAtom] -> [PrimitiveTerm] -> [StdName] -> EpistemicState
initial atoms terms dom = Set.fromList (allWorldStates atoms terms dom)
```

3 Comparing KL and Epistemic Logic

```
module Translator where

import Data.List
import Data.Maybe
import GHC.Num

-- importing syntax and semantics of KL
import SyntaxKL
import SemanticsKL

import qualified Data.Map as Map
import Data.Set (Set)
import qualified Data.Set as Set
```

3.1 Preliminaries

We want to compare KL and Standard Epistemic Logic based on Kripke frames. (Call this SEL). For example, we might want to compare the complexity of model checking for KL and SEL. To do this, we need some way of "translating" between formulas of KL and formulas of SEL, and between KL-models and Kripke frames. This would allow us to, e.g., (1) take a set of KL formulas of various lengths and a set of KL models of various sizes; (2) translate both formulas and models into SEL; (3) do model checking for both (i.e.. on the KL side, and on the SEL side); (4) compare how time/memory scales with length of formula.

Three things to note:

1. The language of KL is predicate logic, plus a knowledge operator. The language of SEL, on the other hand, is propositional logic, plus a knowledge operator. So we can only translate from a fragment of the language of KL to SEL.

2. Kripke models are much more general than KL models. So we can only translate from a fragment of the set of Kripke models into KL models.

3. In Kripke models, there is such a thing as evaluating a formula at a specific world, whereas this has no equivalent in KL models. We need to take this fact into account when thinking about adequacy criteria for a translation function.

3.2 Syntax and Semantics of SEL

The syntax and semantics of SEL is well-known: the language is just the language of basic modal logic, where the Box operator is interpreted as "It is known that...". Models are Kripke models. All this is known from HW2, so we focus on the implementation, here.

3.2.1 Implementation

Syntax

```
data ModForm = P Proposition
             | Neg ModForm
             | Con ModForm ModForm
             | Box ModForm
             | Dia ModForm
             deriving (Eq,Ord,Show)

disj :: ModForm -> ModForm -> ModForm
disj f g = Neg (Con (Neg f) (Neg g))

impl :: ModForm -> ModForm -> ModForm
impl f = disj (Neg f)
```

This is taken from HW2, with one modification: we use "Neg" instead of "Not" as a type constructor for negated modal formula, since "Not" is already being used as a type constructor for KL-formulas (see "SyntaxKL" for details).

Semantics

To represent Kripke Models, we will use the types from HW 2, with a twist: we let the worlds be WorldStates, as defined in "Semantics". This simplifies the translation functions, and doesn't matter mathematically, as the internal constitution of the worlds in a Kripke Model is mathematically irrelevant.

```
--definition of models
type World = WorldState
type Universe = [World]
type Proposition = Int

type Valuation = World -> [Proposition]
type Relation = [(World,World)]

data KripkeModel = KrM
   { universe :: Universe
   , valuation :: Valuation
   , relation :: Relation }

--definition of truth for modal formulas
--truth at a world
makesTrue :: (KripkeModel,World) -> ModForm -> Bool
makesTrue (KrM _ v _, w) (P k)     = k `elem` v w
makesTrue (m,w)          (Neg f)   = not (makesTrue (m,w) f)
makesTrue (m,w)          (Con f g) = makesTrue (m,w) f && makesTrue (m,w) g
makesTrue (KrM u v r, w) (Dia f)   = any (\w' -> makesTrue (KrM u v r, w') f) (r ! w)
makesTrue (KrM u v r, w) (Box f)   = all (\w' -> makesTrue (KrM u v r,w') f) (r ! w)

(!) :: Relation -> World -> [World]
(!) r w = map snd $ filter ((==) w . fst) r

--truth in a model
trueEverywhere :: KripkeModel -> ModForm -> Bool
trueEverywhere (KrM x y z) f = all (\w -> makesTrue (KrM x y z, w) f) x
```

Sometimes, it will still be useful to represent Kripke Models in the old way, using Integers as worlds. We therefore define an additional type IntKripkeModel:

```
type IntWorld = Integer
type IntUniverse = [IntWorld]
type IntValuation = IntWorld -> [Proposition]
type IntRelation = [(IntWorld, IntWorld)]
data IntKripkeModel = IntKrM IntUniverse IntValuation IntRelation
```

Sometimes it will also be usefuel to convert between KripkeModels and IntKripkeModels. To enable this, we provide the following functions:

```
--KripkeModel to IntKripkeModel
translateKrToKrInt :: KripkeModel -> IntKripkeModel
translateKrToKrInt (KrM u v r) = IntKrM u' v' r' where
   ur = nub u -- the function first gets rid of duplicate worlds in the model
   u' = take (length ur) [0..]
   v' n = v (intToWorldState ur n) where
      intToWorldState :: Universe -> Integer -> WorldState
      intToWorldState urc nq = urc !! integerToInt nq
   r' = [(worldStateToInt ur w, worldStateToInt ur w') | (w,w') <- r] where
      worldStateToInt :: Universe -> WorldState -> Integer
      worldStateToInt uni w = toInteger $ fromJust $ elemIndex w uni

convertToWorldStateModel :: IntKripkeModel -> KripkeModel
convertToWorldStateModel (IntKrM intUniv intVal intRel) =
  let worldStates = map makeWorldState intUniv
      worldToInt :: WorldState -> Integer
      worldToInt ws = case find (\(_, w) -> w == ws) (zip intUniv worldStates) of
                        Just (i, _) -> i
                        Nothing -> error "WorldState not found in universe"
      newVal :: Valuation
      newVal ws = intVal (worldToInt ws)
      newRel :: Relation
      newRel = [(makeWorldState i, makeWorldState j) | (i, j) <- intRel]
  in KrM worldStates newVal newRel

makeWorldState :: Integer -> WorldState
makeWorldState n =
  let uniqueAtom = PPred "WorldID" [StdName (show n)]
  in mkWorldState [(uniqueAtom, True)] []
```

To be able to print Models, let's define a Show instance for IntKripkeModels, and for KripkeModels:

```
instance Show IntKripkeModel where
   show (IntKrM uni val rel) = "IntKrM\n" ++ show uni ++ "\n" ++ show [(x, val x) | x <-
       uni ] ++ "\n" ++ show rel
```

We also define a Show Instance for KripkeModels, which just shows the KripkeModel, converted to an IntKripkeModel; the rationale for this is that Integers look much nicer than WorldStates when printed.

```
instance Show KripkeModel where
   show m = show $ translateKrToKrInt m
```

Note that we are breaking with the convention that the show function should return what you need to type into ghci to define the object; however, we feel justified in doing this because of the greater user friendliness it provides.

Later, we will want to compare models for equality; so we'll also define an Eq instance. Comparison for equality will work, at least as long as models are finite. The way this comparison works is by checking that the valuations agree on all worlds in the model. By sorting before checking for equality, we ensure that the order in which worlds appear in the list of worlds

representing the universe, the order in which true propositions at a world appear, and the order in which pairs appear in the relation doesn't affect the comparison.

```
instance Eq KripkeModel where
   (KrM u v r) == (KrM u' v' r') =
      (nub. sort) u == (nub. sort) u' && all (\w -> (nub. sort) (v w) ==  (nub. sort) (v' w
          )) u && (nub. sort) r == (nub. sort) r'

instance Eq IntKripkeModel where
   (IntKrM u v r) == (IntKrM u' v' r') =
      (nub. sort) u == (nub. sort) u' && all (\w -> (nub. sort) (v w) ==  (nub. sort) (v' w
          )) u && (nub. sort) r == (nub. sort) r'
```

NB: the following is possible: Two KripkeModels are equal, we convert both to IntKripkeModels, and the resulting IntKripkeModels are not equal.

Why is this possible? Because when checking for equality of KripkeModels, we ignore the order of worlds in the list that defines the universe; but for the conversion to IntKripkeModels, the order matters!

Similarly, we may get translateModToKr model1 == kripkeM1, but when we print the lhs and rhs of the equation, we get different results. This is perfectly fine, and not unexpected, since printing of KripkeModels works via converting them to IntKripkeModels, and then printing them.

3.3 Translation functions: KL to Kripke

3.3.1 Desiderata

First, what are the types of our translation functions? As mentioned at the beginning, we need to bear several things in mind when deciding this question:

1. The language of KL is predicate logic, plus a knowledge operator. The language of SEL, on the other hand, is propositional logic, plus a knowledge operator. So we can only translate from a fragment of the language of KL to SEL.

2. Kripke models are much more general than KL models. So we can only translate from a fragment of the set of Kripke models into KL models.

3. In Kripke models, there is such a thing as evaluating a formula at a specific world, whereas this has no equivalent in KL models. In fact, evaluating a formula in a KL model is much more closely related to evaluating a formula at a world in a Kripke model, than to evaluating it with respect to a whole Kripke model.

In our implementation, to do justice to the the fact that translation functions can only sensibly be defined for *some* Kripke models, and *some* KL formulas, we use the Maybe monad provided by Haskell.

To do justice to the fact that evaluating in a KL model is more like evaluating a formula at a specific world in a Kripke Model, than like evaluating a formula with respect to a whole Kripke model, we translate from pairs of Kripke models and worlds to KL models, rather than just from Kripke models to KL models.

Thus, these are the types of our translation functions:

1. `translateFormToKr :: Formula -> Maybe ModForm`
2. `translateFormToKr :: Formula -> Maybe ModForm`
3. `translateModToKr :: Model -> KripkeModel`
4. `kripkeToKL :: KripkeModel -> WorldState -> Maybe Model`

What constraints do we want our translation functions to satisfy? We propose that reasonable translation functions should at least satisfy these constraints: for any KL model

`Model w e d`

any translatable Kripke model

`KrM uni val rel`

`g`

1. `Model w e d |= f`

   `(translateModToKr (Model w e d)) w |= fromJust (translateFormToKr f )`

2. `(KrM uni val rel) w |= g`

   `fromJust (kripkeToKL (KrM uni val rel) w) |= translateFormToKL g`

3. `fromJust (translateFormToKL (translateFormToKr f )) = f`
4. `translateFormToKr (fromJust (translateFormToKL g )) = g`
5. `fromJust (kripkeToKL (translateModToKr (Model w e d)) w) = Model w e d`
6. `translateModToKr ( fromJust (kripkeToKL (KrM uni val rel) w)) = KrM uni val rel`

3.3.2 Implementation

Now we get to the implementation of our translation functions.

3.4 Translation functions from $\mathcal{KL}$ to Kripke

Translation functions for formulas:

`translateFormToKr`

subformulas consisting of the predicate letter "P", followed by a standard name by propositional variables; it returns Nothing if the input formula doesn't satisfy this constraint.

```
translateFormToKr :: Formula -> Maybe ModForm
translateFormToKr (Atom (Pred "P" [StdNameTerm (StdName nx)])) = Just $ P (read (drop 1 nx)
    )
translateFormToKr (Not f)                          =  Neg <$> translateFormToKr f
translateFormToKr (Or f g)                         = fmap disj (translateFormToKr f) <*>
    translateFormToKr g
translateFormToKr (K f)                            = Box <$> translateFormToKr f
translateFormToKr _                                = Nothing
```

Translation functions for models:

`translateModToKr`

- the worlds are all the world states in the epistemic state, plus the actual world state;
- for each world, the propositional variables true at it are the translations of the atomic formulas consisting of "P" followed by a standard name that are true at the world state;
- the from within the epistemic state all see each other, and the actual world sees all other worlds.

```
translateModToKr :: Model -> KripkeModel
translateModToKr (Model w e _) = KrM (nub (w:Set.toList e)) val (nub rel) where
   val = trueAtomicPropsAt
   rel = [(v, v') | v <- Set.toList e, v' <- Set.toList e] ++ [(w,v) | v <- Set.toList e]

--the next two are helper functions:

--identifies true atomic formulas at a world that consist of the predicate "P" followed by
    a standard name
trueAtomicPropsAt :: WorldState -> [Proposition]
trueAtomicPropsAt w =
   map (\(Pred "P" [StdNameTerm (StdName nx)]) -> read (drop 1 nx)) trueActualAtoms where
      trueActualAtoms = filter isActuallyAtomic $ map fst (filter snd (Map.toList (
          atomValues w)))

--checks whether an atomic formula consists of the predicate "P" followed by a standard
    name
isActuallyAtomic :: Atom -> Bool
isActuallyAtomic (Pred "P" [StdNameTerm (StdName _)]) = True
isActuallyAtomic _ = False
```

3.5 Translation functions from Kripke to $\mathcal{KL}$

Translation functions for formulas:

`translateFormToKL`

and computes the translated $\mathcal{KL}$ formula. Since SEL is a propositional logic, we will immitate this in the language of $\mathcal{KL}$ by translating it to a unique corresponding atomic formula in $\mathcal{KL}$.

```
-- Translates an SEL formula (propositional modal logic) to a KL formula (predicate logic
    with knowledge operator).
translateFormToKL :: ModForm -> Formula
translateFormToKL (P n) = Atom (Pred "P" [StdNameTerm (StdName ("n" ++ show n))])  -- Maps
    proposition P n to atom P(n), e.g., P 1 -> P(n1)
translateFormToKL (Neg form) = Not (translateFormToKL form)                         --
    Negation is preserved recursively
```

```
translateFormToKL (Con form1 form2) = Not (Or (Not (translateFormToKL form1)) (Not (
    translateFormToKL form2)))  -- Conjunction as   ( f1       f2 )
translateFormToKL (Box form) = K (translateFormToKL form)                          -- Box
     (   ) becomes K, representing knowledge
translateFormToKL (Dia form) = Not (K (Not (translateFormToKL form)))              --
    Diamond (   ) as   K  , representing possibility
```

Translation functions for models:

`kripkeToKL`

and computes a corresponding $\mathcal{KL}$ model which is satisfiability equivalent with the given world in the given model.

KL models (Knowledge Logic models) and Kripke models are frameworks used to represent an agent's knowledge in epistemic logic, but they differ in their structure and approach. A KL model explicitly separates: - An **actual world state**, representing what is true in the real world. - A set of **epistemic world states**, representing the worlds the agent considers possible based on their knowledge.

In contrast, a Kripke model consists of: - A set of possible worlds. - An accessibility relation between worlds, where an agent knows a proposition p at a world w if p is true in all worlds accessible from w.

The task of translating a Kripke model into a KL model involves preserving both: 1. **What is true** in the actual world. 2. **What the agent knows** in that world.

To perform this translation, we take a Kripke model and a specified world w (designated as the actual world) and construct a KL model where: - The **actual world state** corresponds to w. - The **epistemic state** consists of all worlds accessible from w in the Kripke model.

However, not all Kripke models can be straightforwardly translated into KL models while maintaining the semantics of knowledge as defined in KL. In KL, knowledge of a proposition p means that p is true in all world states within the epistemic state. This imposes specific requirements on the structure of the Kripke model's accessibility relation to ensure compatibility with KL's properties of knowledge.

Restrictions on Kripke Models: Transitivity and Euclideanness

In KL, knowledge exhibits introspective properties, such as: - **Positive introspection**: If an agent knows p, then they know that they know p (formally, $Kp \rightarrow KKp$). - **Negative introspection**: If an agent does not know p, then they know that they do not know p (formally, $\neg Kp \rightarrow K\neg Kp$).

These properties imply that the epistemic state must be consistent and stable across all worlds the agent considers possible. Specifically, for any proposition p, the formula KKp (the agent knows that they know p) should hold if and only if Kp (the agent knows p) holds. This equivalence requires the set of worlds in the epistemic state to have a uniform structure.

To capture these introspective properties in a Kripke model, the accessibility relation must satisfy: - **Transitivity**: If world w can access world v (i.e., $w \rightarrow v$), and v can access world u (i.e., $v \rightarrow u$), then w must access u (i.e., $w \rightarrow u$). This ensures that the agent's knowledge extends consistently across all accessible worlds, supporting positive introspection. - **Euclideanness**: If

$w \to v$ and $w \to u$, then $v \to u$. This means that any two worlds accessible from w must be accessible to each other, which is necessary for negative introspection.

When a Kripke model's accessibility relation is both transitive and Euclidean, the set of worlds accessible from w forms an **equivalence class**. In an equivalence class: - Every world is accessible from every other world (including itself, implying reflexivity). - The set is fully connected, meaning the agent's knowledge is uniform across all worlds in the epistemic state.

This fully connected structure aligns with the KL model's epistemic state, where a proposition is known only if it holds in all possible worlds the agent considers. If the worlds accessible from w did not "see each other" (i.e., were not mutually accessible), introspective knowledge claims like $KKp \leftrightarrow Kp$ could not be preserved, as the agent's knowledge would vary inconsistently across the accessible worlds.

The Actual World and the Possibility of Knowing Falsehoods

A key feature of KL models is that the agent can know something false. This means that the actual world state (representing the true state of affairs) does not necessarily belong to the epistemic state (the worlds the agent considers possible). For example, an agent might believe a false proposition, leading them to exclude the actual world from their epistemic state. In the translation from a Kripke model to a KL model: - We designate w as the actual world state. - The epistemic state includes only the worlds accessible from w, which may or may not include w itself.

This flexibility distinguishes KL models from some traditional Kripke-based systems (e.g., S5, where the actual world is typically accessible due to reflexivity), allowing KL to model scenarios where an agent's knowledge deviates from reality.

Conditions for Translation

To successfully translate a Kripke model into a KL model: - The accessibility relation must be **transitive and Euclidean**, ensuring that the worlds accessible from w form an equivalence class. - The translation process then: - Sets the actual world state of the KL model to w. - Defines the epistemic state as the set of all worlds accessible from w in the Kripke model.

If the Kripke model's accessibility relation is not transitive and Euclidean, the translation cannot preserve the introspective properties of knowledge required by KL, and thus it fails. In practice, this restriction is implemented in the function below, which: - Takes a Kripke model and a world w as input. - Verifies that the accessibility relation satisfies transitivity and Euclideanness. - Returns a KL model if the conditions are met, or an indication of failure (e.g., 'Nothing') otherwise.

Example and Intuition

Consider a Kripke model with worlds $\{w, v, u\}$, where: - $w \to v$ and $w \to u$, but $v \not\to u$ (not Euclidean). - At w, the agent knows p if p is true in v and u.

If p is true in v and u, then Kp holds at w. However, for KKp to hold, Kp must be true in both v and u. If $v \not\to u$, then Kp might not hold at v (since v does not access u), breaking the

equivalence $KKp \leftrightarrow Kp$. In a KL model, the epistemic state must ensure mutual accessibility to avoid such inconsistencies, which is why transitivity and Euclideanness are required.

```
-- Main function: Convert Kripke model to KL model
kripkeToKL :: KripkeModel -> WorldState -> Maybe Model
kripkeToKL kr@(KrM univ val rel) w
  | not (isEuclidean kr && isTransitive kr) || not (isInUniv w univ) = Nothing
  | otherwise = Just (Model newWorldState newEpistemicState newDomain)
  where
    -- New actual world based on valuation of w
    newWorldState = createWorldState (val w)

    -- Accessible worlds from w
    accessibleWorlds = [v | (u, v) <- rel, u == w]

    -- New epistemic state: one WorldState per accessible world
    newEpistemicState = Set.fromList [createWorldState (val v) | v <- accessibleWorlds]

    -- Domain (empty for simplicity)
    newDomain = Set.empty

-- Maps an SEL proposition to a KL atom
propToAtom :: Proposition -> Atom
propToAtom n = Pred "P" [StdNameTerm (StdName ("n" ++ show n))]  -- e.g., 1 -> P(n1)

-- Creates a KL WorldState from a list of propositions
createWorldState :: [Proposition] -> WorldState
createWorldState props =
  let atomVals = Map.fromList [(propToAtom p, True) | p <- props]  -- Maps each proposition
       to True
      termVals = Map.empty                                          -- No term valuations
          needed here
  in WorldState atomVals termVals

-- Checks if a world is in the Kripke models universe
isInUniv :: WorldState -> [WorldState] -> Bool
isInUniv = elem  -- Simple membership test

-- Helper: functions as provided
uniqueProps :: ModForm -> [Proposition]
uniqueProps f = nub (propsIn f)
  where
    propsIn (P k)       = [k]
    propsIn (Neg g)     = propsIn g
    propsIn (Con g h)   = propsIn g ++ propsIn h
    propsIn (Box g)     = propsIn g
    propsIn (Dia g)     = propsIn g

-- Generate all possible valuations explicitly
allValuations :: [World] -> [Proposition] -> [Valuation]
allValuations univ props =
  let subsetsP = subsequences props
--      assignToWorlds = sequence (replicate (length univ) subsetsP)
  in [ \w -> let idx = length (takeWhile (/= w) univ)
             in if idx < length univ then assignToWorlds !! idx else []
     | assignToWorlds <- replicate (length univ) subsetsP  ----DRAMA?????
 ]

-- Checks whether a Kripke formula is valid on a given Kripke model
isValidKr :: ModForm -> KripkeModel -> Bool
isValidKr f (KrM univ _ rel) =
  let props = uniqueProps f
      valuations = allValuations univ props
  in all (\v -> all (\w -> makesTrue (KrM univ v rel, w) f) univ) valuations

-- Checks if a Kripke model is Euclidean
isEuclidean :: KripkeModel -> Bool
isEuclidean = isValidKr (disj (Box (Neg (P 1))) (Box (Dia (P 1))))
  --    P1           P1    holds for Euclidean relations

-- Checks if a Kripke model is transitive
isTransitive :: KripkeModel -> Bool
```

```
isTransitive = isValidKr (disj (Neg (Box (P 1))) (Box (Box (P 1))))  --    P1
       P1    holds for transitive relations
```

3.6 Tests

Here are some tests which, for now, you can run simply by typing the name of the test into ghci. They should all return True. They are not yet complete, and after beta, we will integrate testing of the translation functions with testing of the rest of the modules.

```
-- tests for translateFormToKr
formula1, formula2, formula3 :: Formula
formula1 = Atom (Pred "P" [StdNameTerm n1])
formula2 = K (Atom (Pred "P" [StdNameTerm n1]))
formula3 = Not (K (Atom (Pred "P" [StdNameTerm n1])))
-- for these, the translation function should return, respectively:
trFormula1, trFormula2, trFormula3 :: ModForm
trFormula1 = P 1
trFormula2 = Box (P 1)
trFormula3 = Neg (Box (P 1))

ftest1, ftest2, ftest3 :: Bool
ftest1 = fromJust (translateFormToKr formula1) == trFormula1
ftest2 = fromJust (translateFormToKr formula2) == trFormula2
ftest3 = fromJust (translateFormToKr formula3) == trFormula3

--for the next ones, translateFormToKr should return Nothing
form4, form5 :: Formula
form4 = Atom (Pred "Teach" [StdNameTerm n1, StdNameTerm n2])
form5 = Not (K (Atom (Pred "Q" [StdNameTerm n1])))

ftest4, ftest5 :: Bool
ftest4 = isNothing $ translateFormToKr form4
ftest5 = isNothing $ translateFormToKr form5

-- tests for translateModToKr
-- standard name abbreviations:
n1, n2, n3, n4 :: StdName
n1 = StdName "n1" -- ted
n2 = StdName "n2" -- sue
n3 = StdName "n3" -- tina
n4 = StdName "n4" -- tara

-- a model where the actual world is part of the epistemic state
w1, w2, w3, w4 :: WorldState
w1 = mkWorldState [ (PPred "P" [n1], True) ] []
w2 = mkWorldState [ (PPred "P" [n2], True)
                  , (PPred "P" [n3], True) ] []
w3 = mkWorldState [ (PPred "P" [n4], True) ] []
w4 = mkWorldState [] []

e1 :: EpistemicState
e1 = Set.fromList [w1, w2, w3, w4]

domain1 :: Set StdName
domain1 = Set.fromList [n1, n2, n3, n4]

model1 :: Model
model1 = Model w1 e1 domain1

-- if all goes well, this should be converted to the following KripkeModel
kripkeM1 :: KripkeModel
kripkeM1 = KrM uni val rel where
   uni = [w1, w2, w3, w4]
   val world | world == w1 = [1]
             | world == w2 = [2, 3]
             | world == w3 = [4]
             | otherwise   = []
   rel = [(v, v') | v <- uni, v' <- uni]
```

```
test1 :: Bool
test1 = translateModToKr model1 == kripkeM1

-- a model where the actual world is NOT part of the epistemic state
e2 :: EpistemicState
e2 = Set.fromList [w2, w3, w4]

model2 :: Model
model2 = Model w1 e2 domain1

-- if all goes well, this should be converted to the following KripkeModel
kripkeM2 :: KripkeModel
kripkeM2 = KrM uni val rel where
   uni = [w1, w2, w3, w4]
   val world | world == w1 = [1]
             | world == w2 = [2, 3]
             | world == w3 = [4]
             | otherwise   = []
   rel = [(v, v') | v <- es, v' <- es] ++ [(w1, v) | v <- es] where
      es = [w2, w3, w4]

test2 :: Bool
test2 = translateModToKr model2 == kripkeM2

-- a model that contains non-atomic formulas
w2' :: WorldState
w2' = mkWorldState [ (PPred "P" [n2], True)
                   , (PPred "P" [n3], True)
                   , (PPred "R" [n4, n1], True)] []

e1' :: EpistemicState
e1' = Set.fromList [w1, w2', w3, w4]

-- model1' is like model1, except for extra formulas true in w2' that aren't
-- true in w2
model1' :: Model
model1' = Model w1 e1' domain1

-- if all goes well, this should be converted to a model that "looks the same as"
-- the one model1 gets converted to
test3 :: Bool
test3 = translateKrToKrInt (translateModToKr model1) == translateKrToKrInt (
    translateModToKr model1')

-- Add tests for seeing whether the translations interact with their
--     inverses     in the right way
```

3.7 Tests for Translation from Kripke to KL

3.8 Tests for formulae

```
-- | Check if two SEL formulae are logically equivalent across a set of Kripke models
areEquivalent :: [KripkeModel] -> ModForm -> ModForm -> Bool
areEquivalent models f1 f2 =
  all (\m -> all (\w -> makesTrue (m, w) f1 == makesTrue (m, w) f2) (universe m)) models

-- | A small set of predefined Kripke models for testing equivalence
smallModels :: [KripkeModel]
smallModels = [exampleModel1, exampleModel2, exampleModel3, exampleModel4,
               exampleModel5, exampleModel6, exampleModel7]

-- Test translation of atomic proposition
testAtomic :: Bool
testAtomic = let klForm = translateFormToKL (P 1)
                 expected = Atom (Pred "P" [StdNameTerm (StdName "n1")])
             in klForm == expected
```

```haskell
-- Test translation of negation
testNegation :: Bool
testNegation = let g = Neg (P 1)
                   klForm = translateFormToKL g
                   expected = Not (Atom (Pred "P" [StdNameTerm (StdName "n1")]))
               in klForm == expected

-- Test translation of conjunction
testConjunction :: Bool
testConjunction = let g = Con (P 1) (P 2)
                      klForm = translateFormToKL g
                      expected = Not (Or (Not (Atom (Pred "P" [StdNameTerm (StdName "n1")]))
                          ))
                                         (Not (Atom (Pred "P" [StdNameTerm (StdName "n2")])))
                                             ))
                  in klForm == expected

-- Test translation of box operator
testBox :: Bool
testBox = let g = Box (P 1)
              klForm = translateFormToKL g
              expected = K (Atom (Pred "P" [StdNameTerm (StdName "n1")]))
          in klForm == expected

-- Test translation of diamond operator
testDiamond :: Bool
testDiamond = let g = Dia (P 1)
                  klForm = translateFormToKL g
                  expected = Not (K (Not (Atom (Pred "P" [StdNameTerm (StdName "n1")]))))
              in klForm == expected

-- Test invertibility of a simple formula using logical equivalence
testFormInvertSimple :: Bool
testFormInvertSimple =
  let g = Box (P 1)                       -- Original formula:   P1
      klForm = translateFormToKL g        -- Translate to KL
      selForm = fromJust (translateFormToKr klForm)  -- Back-translate to SEL
  in areEquivalent smallModels g selForm  -- Check equivalence

-- Test invertibility of a complex formula using logical equivalence
testFormInvertComplex :: Bool
testFormInvertComplex =
  let g = Box (Con (P 1) (Neg (P 2)))   -- Original formula:    (P1      P2 )
      klForm = translateFormToKL g       -- Translate to KL
      selForm = fromJust (translateFormToKr klForm)  -- Back-translate to SEL
  in areEquivalent smallModels g selForm  -- Check equivalence

-- Aggregate all formula tests with diagnostic output
testAllFormulae :: String
testAllFormulae =
  let results = [ ("testAtomic", testAtomic)
                , ("testNegation", testNegation)
                , ("testConjunction", testConjunction)
                , ("testBox", testBox)
                , ("testDiamond", testDiamond)
                , ("testFormInvertSimple", testFormInvertSimple)
                , ("testFormInvertComplex", testFormInvertComplex)
                ]
      failures = [name | (name, result) <- results, not result]
  in if null failures
     then "All formula tests passed"
     else "Failed formula tests: " ++ unwords failures
```

3.9 Tests for Models

```haskell
-- | Simplified check if two Kripke models are bisimilar.
areBisimilar :: KripkeModel -> KripkeModel -> Bool
areBisimilar km1 km2 =
  let univ1 = universe km1
      univ2 = universe km2
```

```
      rel1 = relation km1
      rel2 = relation km2
      val1 = valuation km1
      val2 = valuation km2
      initialRel = [(w1, w2) | w1a <- univ1, w2a <- univ2, val1 w1a == val2 w2a]
      satisfiesBackForth r (w1b, w2b) =
        all (\v1 -> any (\v2 -> (v1, v2) `elem` r) (successors w2b rel2))
            (successors w1b rel1) &&
        all (\v2 -> any (\v1 -> (v1, v2) `elem` r) (successors w1b rel1))
            (successors w2b rel2)
      largestBisimulation = until (\r -> r == filter (satisfiesBackForth r) r)
                                  (\r -> filter (satisfiesBackForth r) r)
                                  initialRel
  in not (null largestBisimulation) &&
     all (\w1c -> any (\(w1', _) -> w1c == w1') largestBisimulation) univ1 &&
     all (\w2c -> any (\(_, w2'd) -> w2c == w2'd) largestBisimulation) univ2

-- | Get successor worlds in a relation
successors :: WorldState -> [(WorldState, WorldState)] -> [WorldState]
successors w rel = [v | (u, v) <- rel, u == w]

-- Helper function to test truth preservation
testTruthPres :: KripkeModel -> WorldState -> ModForm -> Bool
testTruthPres km w g =
  case kripkeToKL km w of
    Nothing -> False  -- Translation failed, so truth not preserved
    Just klModel -> let klFormula = translateFormToKL g
                    in makesTrue (km, w) g == satisfiesModel klModel klFormula

-- Test translation succeeds for S5 model
testTranslationSucceedsModel3 :: Bool
testTranslationSucceedsModel3 = isJust (kripkeToKL exampleModel3 (makeWorldState 30))

-- Test translation succeeds for clustered model
testTranslationSucceedsModel6 :: Bool
testTranslationSucceedsModel6 = isJust (kripkeToKL exampleModel6 (makeWorldState 60))

-- Test translation fails for linear non-transitive model
testTranslationFailsModel2 :: Bool
testTranslationFailsModel2 = isNothing (kripkeToKL exampleModel2 (makeWorldState 20))

-- Test translation fails for cyclic non-transitive model
testTranslationFailsModel5 :: Bool
testTranslationFailsModel5 = isNothing (kripkeToKL exampleModel5 (makeWorldState 50))

-- Test truth preservation for atomic proposition in S5 model
testTruthPresP1Model3 :: Bool
testTruthPresP1Model3 = testTruthPres exampleModel3 (makeWorldState 30) (P 1)

-- Test truth preservation for box operator in S5 model
testTruthPresBoxP1Model3 :: Bool
testTruthPresBoxP1Model3 = testTruthPres exampleModel3 (makeWorldState 30) (Box (P 1))

-- Test truth preservation for diamond operator in S5 model
testTruthPresDiaP1Model3 :: Bool
testTruthPresDiaP1Model3 = testTruthPres exampleModel3 (makeWorldState 30) (Dia (P 1))

-- Test truth preservation for conjunction in clustered model
testTruthPresConModel6 :: Bool
testTruthPresConModel6 = testTruthPres exampleModel6 (makeWorldState 60) (Con (P 1) (P 2))

-- Test model invertibility for S5 model using bisimulation
testModelInvertModel3 :: Bool
testModelInvertModel3 =
  let km = exampleModel3                -- Original S5 Kripke model
      w = makeWorldState 30             -- A starting world state
      klm = fromJust (kripkeToKL km w)  -- Translate to KL
      kmBack = translateModToKr klm     -- Back-translate to Kripke model
  in areBisimilar kmBack km             -- Check bisimulation

-- Test contradiction is false in modelKL7
testContradictionFalseModel7 :: Bool
```

```
testContradictionFalseModel7 = not (checkModel modelKL7 (translateFormToKL (Con (P 2) (Neg
    (P 2)))))


-- Test reflexivity axiom false in modelKL7 (w70 not reflexive)
testRefFalseModel7 :: Bool
testRefFalseModel7 = checkModel modelKL7 ref


-- Test reflexivity axiom true in modelKL7b (w71 in cluster)
testRefTrueModel7b :: Bool
testRefTrueModel7b = checkModel modelKL7b ref

-- Aggregate all model tests with diagnostic output
testAllModels :: String
testAllModels =
  let results = [ ("testTranslationSucceedsModel3", testTranslationSucceedsModel3)
                , ("testTranslationSucceedsModel6", testTranslationSucceedsModel6)
                , ("testTranslationFailsModel2", testTranslationFailsModel2)
                , ("testTranslationFailsModel5", testTranslationFailsModel5)
                , ("testTruthPresP1Model3", testTruthPresP1Model3)
                , ("testTruthPresBoxP1Model3", testTruthPresBoxP1Model3)
                , ("testTruthPresDiaP1Model3", testTruthPresDiaP1Model3)
                , ("testTruthPresConModel6", testTruthPresConModel6)
                , ("testModelInvertModel3", testModelInvertModel3)
                , ("testContradictionFalseModel7", testContradictionFalseModel7)
                , ("testRefFalseModel7", testRefFalseModel7)
                , ("testRefTrueModel7b", testRefTrueModel7b)
                ]
      failures = [name | (name, result) <- results, not result]
  in if null failures
     then "All model tests passed"
     else "Failed model tests: " ++ unwords failures
```

3.10 example models used to test the translations from Kripke to KL

```
--IntegerKripkeModel to KripkeModel

--Example 1 : Reflexive, isolated worlds
intW0, intW1, intW2 :: IntWorld
intW0 = 0
intW1 = 1
intW2 = 2

intUniverse1 :: IntUniverse
intUniverse1 = [intW0, intW1, intW2]

intValuation1 :: IntValuation
intValuation1 w
  | w == 0 || w == 1 = [1]  -- Proposition 1 holds in worlds 0 and 1
  | w == 2           = []   -- No propositions hold in world 2
  | otherwise        = []

intRelation1 :: IntRelation
intRelation1 = [(0, 0), (0, 1), (1, 0), (1, 1), (2, 2)]

intModel1 :: IntKripkeModel
intModel1 = IntKrM intUniverse1 intValuation1 intRelation1

-- Convert to WorldState-based model
exampleModel1 :: KripkeModel
exampleModel1 = convertToWorldStateModel intModel1

--Example2 : Linear
-- Integer-based model
intW20, intW21, intW22 :: IntWorld
intW20 = 20
intW21 = 21
intW22 = 22
```

```
intUniverse2 :: IntUniverse
intUniverse2 = [intW20, intW21, intW22]

intValuation2 :: IntValuation
intValuation2 w
  | w == 20 = [1]  -- Proposition 1 holds in world 20
  | w == 21 = []   -- No propositions in world 21
  | w == 22 = [1]  -- Proposition 1 holds in world 22
  | otherwise = []

intRelation2 :: IntRelation
intRelation2 = [(20, 21), (21, 22)]  -- Linear: 20 -> 21 -> 22

intModel2 :: IntKripkeModel
intModel2 = IntKrM intUniverse2 intValuation2 intRelation2

exampleModel2 :: KripkeModel
exampleModel2 = convertToWorldStateModel intModel2

--Example3 : S5
-- Integer-based model
intW30, intW31, intW32 :: IntWorld
intW30 = 30
intW31 = 31
intW32 = 32

intUniverse3 :: IntUniverse
intUniverse3 = [intW30, intW31, intW32]

intValuation3 :: IntValuation
intValuation3 w
  | w == 30 = [1]  -- Proposition 1 holds in world 30
  | otherwise = [] -- No propositions elsewhere

intRelation3 :: IntRelation
intRelation3 = [(w1e, w2e) | w1e <- intUniverse3, w2e <- intUniverse3]  -- Fully connected

intModel3 :: IntKripkeModel
intModel3 = IntKrM intUniverse3 intValuation3 intRelation3

exampleModel3 :: KripkeModel
exampleModel3 = convertToWorldStateModel intModel3

--Example 4 : Empty relation
-- Integer-based model
intW40, intW41, intW42 :: IntWorld
intW40 = 40
intW41 = 41
intW42 = 42

intUniverse4 :: IntUniverse
intUniverse4 = [intW40, intW41, intW42]

intValuation4 :: IntValuation
intValuation4 w
  | w == 40 = [1]  -- Proposition 1 holds in world 40
  | w == 41 = []   -- No propositions in world 41
  | w == 42 = [2]  -- Proposition 2 holds in world 42
  | otherwise = []

intRelation4 :: IntRelation
intRelation4 = []  -- Empty relation

intModel4 :: IntKripkeModel
intModel4 = IntKrM intUniverse4 intValuation4 intRelation4

exampleModel4 :: KripkeModel
exampleModel4 = convertToWorldStateModel intModel4

--Example5 : Cyclic with multiple propositions
-- Integer-based model
intW50, intW51, intW52 :: IntWorld
intW50 = 50
```

```
intW51 = 51
intW52 = 52

intUniverse5 :: IntUniverse
intUniverse5 = [intW50, intW51, intW52]

intValuation5 :: IntValuation
intValuation5 w
  | w == 50 = [1, 2]  -- Propositions 1 and 2 hold in world 50
  | w == 51 = [1]     -- Proposition 1 holds in world 51
  | w == 52 = []      -- No propositions in world 52
  | otherwise = []

intRelation5 :: IntRelation
intRelation5 = [(50, 51), (51, 52), (52, 50)]  -- Cycle: 50 -> 51 -> 52 -> 50

intModel5 :: IntKripkeModel
intModel5 = IntKrM intUniverse5 intValuation5 intRelation5

exampleModel5 :: KripkeModel
exampleModel5 = convertToWorldStateModel intModel5

--Example 6 : Euclidean, Transitive, and Reflexive Frame
-- Integer-based model
intW60, intW61, intW62, intW63, intW64 :: IntWorld
intW60 = 60
intW61 = 61
intW62 = 62
intW63 = 63
intW64 = 64

intUniverse6 :: IntUniverse
intUniverse6 = [intW60, intW61, intW62, intW63, intW64]

intValuation6 :: IntValuation
intValuation6 w
  | w == 60 = [1]        -- Proposition 1 holds in world 60
  | w == 61 = [1, 2]     -- Propositions 1 and 2 hold in world 61
  | w == 62 = []         -- No propositions in world 62
  | w == 63 = [2]        -- Proposition 2 holds in world 63
  | w == 64 = [1, 2]     -- Propositions 1 and 2 hold in world 64
  | otherwise = []

intRelation6 :: IntRelation
intRelation6 = let cluster1 = [60, 61, 62]  -- First cluster: fully connected
                   cluster2 = [63, 64]      -- Second cluster: fully connected
               in [(w1f, w2f) | w1f <- cluster1, w2f <- cluster1] ++
                  [(w1f, w2f) | w1f <- cluster2, w2f <- cluster2]

intModel6 :: IntKripkeModel
intModel6 = IntKrM intUniverse6 intValuation6 intRelation6

exampleModel6 :: KripkeModel
exampleModel6 = convertToWorldStateModel intModel6


--Example 7: Euclidean, Transitive, but not Reflexive Frame
-- Integer-based model
intW70, intW71, intW72, intW73, intW74, intW75 :: IntWorld
intW70 = 70
intW71 = 71
intW72 = 72
intW73 = 73
intW74 = 74
intW75 = 75

intUniverse7 :: IntUniverse
intUniverse7 = [intW70, intW71, intW72, intW73, intW74, intW75]

intValuation7 :: IntValuation
intValuation7 w
  | w == 70 = [1]
  | w == 71 = [1, 2]
```

```
  | w == 72 = [2]
  | w == 73 = []
  | w == 74 = [1]
  | w == 75 = [1, 2]
  | otherwise = []

intRelation7 :: IntRelation
intRelation7 = let cluster = [71, 72, 73, 74, 75]
               in [(70, w) | w <- cluster] ++ [(w1g, w2g) | w1g <- cluster, w2g <- cluster]

intModel7 :: IntKripkeModel
intModel7 = IntKrM intUniverse7 intValuation7 intRelation7

exampleModel7 :: KripkeModel
exampleModel7 = convertToWorldStateModel intModel7

-- Define modelKL7 and modelKL7b from exampleModel7
modelKL7 :: Model
modelKL7 = fromJust (kripkeToKL exampleModel7 (makeWorldState 70))

modelKL7b :: Model
modelKL7b = fromJust (kripkeToKL exampleModel7 (makeWorldState 71))

-- Define ref as the reflexivity axiom K P(n1) -> P(n1)
ref :: Formula
ref = translateFormToKL (impl (Box (P 99)) (P 99))
```

4 Tableau-Based Satisfiability and Validity Checking in $\mathcal{KL}$

Note: For the Beta-version, we omitted function symbol evaluation, limiting the satisfiability and validity checking to a propositional-like subset.
This subsection implements satisfiability and validity checkers for $\mathcal{KL}$ using the tableau method, a systematic proof technique that constructs a tree to test formula satisfiability by decomposing logical components and exploring possible models. In $\mathcal{KL}$, this requires handling both first-order logic constructs (quantifiers, predicates) and the epistemic operator K, which requires tracking possible worlds. Note that the full first-order epistemic logic with infinite domains is in general undecidable (Levesque and Lakemeyer 2001 p. 173), so we adopt a semi-decision procedure: it terminates with "satisfiable" if an open branch is found but may loop infinitely for unsatisfiable cases due to the infinite domain $\mathcal{N}$. The Tableau module builds on SyntaxKL and SemanticsKL:

```
module Tableau where

import SyntaxKL
import SemanticsKL
import Data.Set (Set)
import qualified Data.Set as Set
```

Tableau Approach
The tableau method tests satisfiability as follows: A formula α is satisfiable if there exists an epistemic state e and a world $w \in e$ such that $e, w \models \alpha$. The tableau starts with α and expands it, seeking an open (non-contradictory) branch representing a model. A formula α is valid if it holds in all possible models ($e, w \models \alpha$ for all e, w). We test validity by checking if $\neg\alpha$ unsatisfiable (i.e., all tableau branches close). For $\mathcal{KL}$ we have to handle two things:

- Infinite domains: $\mathcal{KL}$ assumes a countably infinite set of standard names (Levesque and Lakemeyer 2001, p.23). The tableau method handles this via parameters (free variables) and δ-rules (existential instantiation), introducing new names as needed. This means that

we Use a countably infinite supply of parameters (e.g., a1, a2,...) instead of enumerating all standard names.

- Modal handling: The K-operator requires branching over possible worlds within an epistemic state.

First, we define new types for the tableau node and branch: Nodes pair formulas with world identifiers, and branches track nodes and used parameters.

```
-- A tableau node: formula labeled with a world
data Node = Node Formula World deriving (Eq, Show)

type World = Int     -- World identifier (0, 1, ...)

-- A tableau branch: list of nodes and set of used parameters
data Branch = Branch { nodes :: [Node], params :: Set StdName } deriving (Show)
```

Tableau Rules

Rules decompose formulas, producing either a closed branch (contradictory) or open branches (consistent). applyRule implements these rules, handling logical and epistemic operators. The rules are applied iteratively to unexpanded nodes until all branches are either closed or fully expanded (open).

```
-- Result of applying a tableau rule
data RuleResult = Closed | Open [Branch] deriving (Show)

-- Generates fresh parameters not in the used set
newParams :: Set StdName -> [StdName]
newParams used = [StdName ("a" ++ show i) | i <- [(1::Int)..], StdName ("a" ++ show i) `Set
    .notMember` used]

-- Applies tableau rules to a node on a branch
applyRule :: Node -> Branch -> RuleResult
applyRule (Node f w) branch = case f of
  Atom _ -> Open [branch]   -- If formula is an atom: Do nothing; keep the formula in the
      branch.
  Not (Atom _) -> Open [branch]  -- Negated atoms remain, checked by isClosed
  Equal _ _ -> Open [branch]  -- Keep equality as is; closure checks congruence
  Not (Equal _ _) -> Open [branch]  -- Keep negated equality
  Not (Not f') -> Open [Branch (Node f' w : nodes branch) (params branch)] -- Case: double
      negation, e.g., replace $\neg \neg \varphi$ with $\varphi$
  Not (Or f1 f2) -> Open [Branch (Node (Not f1) w : Node (Not f2) w : nodes branch) (params
       branch)] -- Case: negated disjunction
  Not (Exists x f') -> Open [Branch (Node (klforall x (Not f')) w : nodes branch) (params
      branch)] -- Case:: negated existential
  Not (K f') -> Open [expandKNot f' w branch] -- Case: negated knowledge
  Or f1 f2 -> Open [ Branch (Node f1 w : nodes branch) (params branch)
                   , Branch (Node f2 w : nodes branch) (params branch) ] -- Disjunction
                       rule, split the branch
  Exists x f' ->   -- Existential rule ($\delta$-rule), introduce a fresh parameter a (e.g
      .,  a 1  ) not used elsewhere, substitute x with a, and continue
    let newParam = head (newParams (params branch))
        newBranch = Branch (Node (subst x newParam f') w : nodes branch)
                           (Set.insert newParam (params branch))
    in Open [newBranch]
  K f' -> Open [expandK f' w branch] -- Knowledge rule, add formula to a new world

-- Expands formula K \varphi to a new world
expandK :: Formula -> World -> Branch -> Branch
expandK f w branch = Branch (Node f (w + 1) : nodes branch) (params branch)

-- Expands \not K \varphi to a new world
expandKNot :: Formula -> World -> Branch -> Branch
expandKNot f w branch = Branch (Node (Not f) (w + 1) : nodes branch) (params branch)
```

Branch Closure

isClosed determines whether a tableau branch is contradictory (closed) or consistent (open).

A branch closes if it contains an explicit contradiction, meaning no model can satisfy all the formulas in that branch. If a branch is not closed, it is potentially part of a satisfiable interpretation. The input is a Branch, which has nodes :: [Node] (each Node f w is a formula f in world w) and params :: Set StdName (used parameters). The function works as follows: first, we collect the atoms ((a, w, True) for positive atoms (Node (Atom a) w); (a, w, False) for negated Atoms (Node (Not (Atom a)) w)). For example, if nodes = [Node (Atom P(n1)) 0, Node (Not (Atom P(n1))) 0], then atoms = [(P(n1), 0, True), (P(n1), 0, False)]. Next, we collect the equalities. After this, we check the atom contradictions. There we use *any* to find pairs in *atoms* and return True if a contradiction exists. In a subsequent step, we check for equality contradictions. The result of the function is atomContra || eqContra: this is True if either type of contradiction is found and False otherwise. This function reflects the semantic requirement that a world state w in an epistemic state e can not assign both True and False to the same ground atom or equality

```
-- Branch closure with function symbols
isClosed :: Branch -> Bool
isClosed b =
  let atoms = [(a, w, True) | Node (Atom a) w <- nodes b]
            ++ [(a, w, False) | Node (Not (Atom a)) w <- nodes b]
      equals = [((t1, t2), w, True) | Node (Equal t1 t2) w <- nodes b]
             ++ [((t1, t2), w, False) | Node (Not (Equal t1 t2)) w <- nodes b]
      atomContra = any (\(a1, w1, b1) -> any (\(a2, w2, b2) -> a1 == a2 && w1 == w2 && b1
          /= b2) atoms) atoms
      eqContra = any (\((t1, t2), w1, b1) -> any (\((t3, t4), w2, b2) ->
                   t1 == t3 && t2 == t4 && w1 == w2 && b1 /= b2) equals) equals
  in atomContra || eqContra  -- True if any contradiction exists
```

Tableau Expasion

Next, we have the function expandTableau. expandTableau iteratively applies tableau rules to expand all branches, determining if any remain open (indicating satisfiability). It returns Just branches if at least one branch is fully expanded and open, and Nothing if all branches close. This function uses recursion. It continues until either all branches are closed or some are fully expanded

```
-- Expands the tableau, returning open branches if satisfiable
expandTableau :: [Branch] -> Maybe [Branch]
expandTableau branches
  | all isClosed branches = Nothing --If every branch is contradictory, return Nothing
  | any (null . nodes) branches = Just branches --If any branch has no nodes left to expand
       (and isn't closed), it's open and complete
  | otherwise = do
      let (toExpand, rest) = splitAt 1 branches --Take the first branch (toExpand) and
          leave the rest.
          branch = head toExpand --Focus on this branch.
          node = head (nodes branch) --Pick the first unexpanded node.
          remaining = Branch (tail (nodes branch)) (params branch) --he branch minus the
              node being expanded.
      case applyRule node remaining of
        Closed -> expandTableau rest --Skip this branch, recurse on rest.
        Open newBranches -> expandTableau (newBranches ++ rest) --Add the new branches (e.g
            ., from \lor or \exists) to rest, recurse.
```

Top-Level Checkers

As top-level function we use isSatisfiable and isValid. isSatisfiable tests whether a formula f has a satisfying model. It starts the tableau process and interprets the result. This function gets a Formula f as an input and then creates a single branch with Node f 0 (formula f in world 0) and an empty set of parameters. Next, it calls expandTablaeu on this initial branch. It then interprets the result: if expandTableau returns Just _, this means, that at least one open branch exists, thus, the formula is satisfiable. If expandTableau returns Nothing, this means that all

branches are closed and the formula is unsatisfiable.

```
-- Tests if a formula is satisfiable
isSatisfiable :: Formula -> Bool
isSatisfiable f = case expandTableau [Branch [Node f 0] Set.empty] of
  Just _ -> True
  Nothing -> False
```

The three function isSatisfiable, expandTableau, and isClosed interact as follows: isSatisfiable starts the process with a single branch containing the formula. expandTableau recursively applies applyRule to decompose formulas, creating new branches as needed (e.g., for $\vee$, $\exists$). isClosed checks each branch for contradictions, guiding expandTableau to prune closed branches or halt with an open one.

```
-- Tests if a formula is valid
isValid :: Formula -> Bool
isValid f = not (isSatisfiable (Not f))
```

5 Semantics Tests

```
{-# LANGUAGE InstanceSigs #-}

module Main where

import SemanticsKL
import SyntaxKL
import Generators

import qualified Data.Map as Map
import qualified Data.Set as Set

import Test.Hspec
import Test.QuickCheck
import Control.Exception (evaluate)
```

The following tests are for the semantics of $\mathcal{KL}$, which are defined in the SemanticsKL module. The tests are written using the Hspec testing framework and QuickCheck for property-based testing. The tests cover the evaluation of terms, formulas, and models, as well as model checking function. The Generators file provides helper functions for generating implementing testing, but have been omitted for brevity.

```
main :: IO ()
main = hspec $ do
    describe "Eq is derived for the relevant KL Semantics" $ do
        it "WorldState Eq is derived" $ do
            property $ \w -> w == (w :: WorldState)
        -- TODO: add test to be sure that the world state only has primitive terms and
            atoms
    describe "evalTerm" $ do
        it "evalTerm errors with variables" $ do
            property $ \x -> evaluate (evalTerm (WorldState Map.empty Map.empty) (VarTerm x
                )) `shouldThrow` anyException
        it "evalTerm returns the StdName for StdNameTerm" $ do
            property $ \w n -> evalTerm w (StdNameTerm n) == n
        it "evalTerm returns the StdName after applying all functions (depth 2)" $ do
            let n1 = StdName "n1"
                n2 = StdName "n2"
                n3 = StdName "n3"
                n4 = StdName "n4"
                w = WorldState Map.empty (Map.fromList [
                                        (FuncAppTerm "f" [StdNameTerm n1, StdNameTerm
                                            n2],  n3),
                                        (FuncAppTerm "g" [StdNameTerm n4], n1)
```

```
                                    ])
            t = FuncAppTerm "f" [FuncAppTerm "g" [StdNameTerm n4], StdNameTerm n2]
        evalTerm w t `shouldBe` StdName "n3"
describe "isGround" $ do
    it "isGround returns False for VarTerm" $ do
        property $ \n -> not $ isGround (VarTerm n)
    it "isGround returns True for GroundFuncAppTerm or GroundStdNameTerm" $ do
       property $ forAll genGroundTerm $ \t -> isGround (t :: Term)
    it "isGround returns False for complex FuncAppTerm with a non-ground argument" $ do
        let term = FuncAppTerm "f" [FuncAppTerm "g" [VarTerm $ Var "x", StdNameTerm $
            StdName "n1"]]
        isGround term `shouldBe` False
describe "isGroundFormula" $ do
    it "isGroundFormula returns True for groundFormula" $ do
        property $ forAll genGroundFormula $ \f -> isGroundFormula (f :: Formula)
    it "isGroundFormula returns False for Exists" $ do
        property $ \n f -> not $ isGroundFormula (Exists (Var n) (f :: Formula))
    it "isGroundFormula returns False for Atom with a non-ground term" $ do
        isGroundFormula (Atom (Pred "P" [VarTerm $ Var "x"])) `shouldBe` False
    it "isGroundFormula returns False for Equal with a non-ground term" $ do
        isGroundFormula (Equal (VarTerm $ Var"x") (StdNameTerm $ StdName "n1")) `
            shouldBe` False
describe "substTerm" $ do
    it "substTerm replaces the variable with the StdName" $ do
        let term = FuncAppTerm "f" [VarTerm $ Var "x", StdNameTerm $ StdName "n1"]
        substTerm (Var "x") (StdName "n2") term `shouldBe` FuncAppTerm "f" [StdNameTerm
             $ StdName "n2", StdNameTerm $ StdName "n1"]
    it "substTerm does not replace the wrong variable" $ do
        let term = FuncAppTerm "f" [VarTerm $ Var "y", StdNameTerm $ StdName "n1"]
        substTerm (Var "x") (StdName "n2") term `shouldBe` term
describe "subst" $ do
    it "subst replaces the variable with the StdName in an Atom" $ do
        let atom = Atom (Pred "P" [VarTerm $ Var "x"])
        show (subst (Var "x") (StdName "n1") atom) `shouldBe` show (Atom (Pred "P" [
            StdNameTerm $ StdName "n1"]))
    it "subst replaces the variable with the StdName in an Equal" $ do
        let formula = Equal (VarTerm $ Var "x") (StdNameTerm $ StdName "n1")
        show (subst (Var "x") (StdName "n2") formula) `shouldBe` show (Equal (
            StdNameTerm $ StdName "n2") (StdNameTerm $ StdName "n1"))
    it "subst replaces the variable with the StdName in a Not formula" $ do
        let formula = Not (Atom (Pred "P" [VarTerm $ Var "x"]))
        show (subst (Var "x") (StdName "n1") formula) `shouldBe` show (Not (Atom (Pred
            "P" [StdNameTerm $ StdName "n1"])))
    it "subst replaces the variable with the StdName in an Or formula" $ do
        let formula = Or (Atom (Pred "P" [VarTerm $ Var "x"])) (Atom (Pred "Q" [VarTerm
             $ Var "y"]))
        show (subst (Var "x") (StdName "n1") formula) `shouldBe` show (Or (Atom (Pred "
            P" [StdNameTerm $ StdName "n1"])) (Atom (Pred "Q" [VarTerm $ Var "y"])))
    it "subst replaces the variable with the StdName in an Exists if the variable not
        in Exists scope" $ do
        let formula = Exists (Var "x") (Atom (Pred "P" [VarTerm $ Var "x", VarTerm $
            Var "y"]))
        -- replaces y with n2
        show (subst (Var "y") (StdName "n2") formula) `shouldBe` show (Exists (Var "x")
             (Atom (Pred "P" [VarTerm $ Var "x", StdNameTerm $ StdName "n2"])))
    it "subst does not replace the variable with the StdName in Exists if the variable
        is in the Exists scope" $ do
        let formula = Exists (Var "x") (Atom (Pred "P" [VarTerm $ Var "x", VarTerm $
            Var "y"]))
        -- does not replace x with n2
        show (subst (Var "x") (StdName "n2") formula) `shouldBe` show formula
    it "subst replaces the variable with the StdName in a K formula" $ do
        let formula = K (Atom (Pred "P" [VarTerm $ Var "x"]))
        show (subst (Var "x") (StdName "n2") formula) `shouldBe` show (K (Atom (Pred "P
            " [StdNameTerm $ StdName "n2"])))
describe "satisfiesModel" $ do
    let x = Var "x"
        n1 = StdNameTerm $ StdName "n1"
        n2 = StdNameTerm $ StdName "n2"
        p = Atom (Pred "P" [])
        px = Atom (Pred "P" [VarTerm x])
        py = Atom (Pred "P" [VarTerm $ Var "y"])
        pt = Atom (Pred "P" [n1])
```

```
    context "satisfiesModel satisfies validities when atoms are ground" $ do
        it "satisfiesModel satisfies P -> ~~ P" $ do
            property $ \m -> satisfiesModel m (Or (Not p) (Not (Not p))) `shouldBe`
                True
        it "satisfiesModel satisfies P(t) -> ~~ P(t)" $ do
            property $ \m -> satisfiesModel m (Or (Not pt) (Not (Not pt))) `shouldBe`
                True
        it "satisfiesModel errors for P(x) -> ~~ P(x)" $ do
            property $ \m -> evaluate (satisfiesModel m (Or (Not px) (Not (Not px)))) `
                shouldThrow` anyException
        it "satisfiesModel satisfies t=t" $ do
            property $ \m -> satisfiesModel m (Equal n1 n1) `shouldBe` True
        it "satisfiesModel errors for x=x" $ do
            property $ \m -> evaluate (satisfiesModel m (Equal (VarTerm x) (VarTerm x))
                ) `shouldThrow` anyException
        it "satisfiesModel satisfies ForAll x (P(x) -> P(x))" $ do
            property $ \m -> satisfiesModel m (Not (Exists x (Not (Or (Not px) px)))) `
                shouldBe` True
        it "satisfiesModel satisfies ForALL x (P(x) -> ~~ P(x))" $ do
            property $ \m -> satisfiesModel m (Not (Exists x (Not (Or (Not px) (Not (
                Not px))))) ) `shouldBe` True
        it "satisfiesModel satisfies ForAll x (P(x) -> Exists y P(y))" $ do
            property $ \m -> satisfiesModel m (Not (Exists x (Not (Or (Not px) (Exists
                (Var "y") py)) ))) `shouldBe` True
        it "satisfiesModel satisfies ((n1 = n2) -> K (n1 = n2)" $ do
            property $ \m -> satisfiesModel m (Or (Not (Equal n1 n2)) (K (Equal n1 n2))
                ) `shouldBe` True
        it "satisfiesModel satisfies ((n1 /= n2) -> K (n1 /= n2)" $ do
            property $ \m -> satisfiesModel m (Or (Not (Not (Equal n1 n2))) (K (Not (
                Equal n1 n2)))) `shouldBe` True
        it "satisfiesModel satisfies (K alpha -> K K alpha)" $ do
            property $ \m -> satisfiesModel m (Or (Not (K pt)) (K (K pt))) `shouldBe`
                True
        it "satisfiesModel satisfies (~K alpha -> K ~K alpha)" $ do
            property $ \m -> satisfiesModel m (Or (Not (Not (K pt))) (K (Not (K pt))))
                `shouldBe` True
    context "satisfiesModel does not satisfy contradictions when atoms are ground" $ do
        it "satisfiesModel does not satisfy ~(P v ~P)" $ do
            property $ \m -> satisfiesModel m (Not (Or p (Not p))) `shouldBe` False
        it "satisfiesModel does not satisfy (Exists x (x /= x))" $ do
            property $ \m -> satisfiesModel m (Exists x (Not (Equal (VarTerm x) (
                VarTerm x)))) `shouldBe` False
describe "freeVars" $ do
        let x = Var "x"
            y = Var "y"
            n1 = StdNameTerm $ StdName "n1"
            n2 = StdNameTerm $ StdName "n2"
            px = Atom (Pred "P" [VarTerm x])
            py = Atom (Pred "P" [VarTerm y])
            pf = Atom (Pred "P" [FuncAppTerm "f" [VarTerm x], FuncAppTerm "g" [VarTerm
                y]])
        it "freeVars returns nothing if no free var in formula" $ do
            let f = Exists x (Or (Or (Not px) (Exists y py)) (Equal n1 n2))
            freeVars f `shouldBe` Set.fromList []
        it "freeVars returns the free variables in a simple formula" $ do
            let f = Or (Or (Not px) py) (Equal n1 n2)
            freeVars f `shouldBe` Set.fromList [x, y]
        it "freeVars returns the free variables in a complex formula" $ do
            let f = Exists x (Or (Or (Not px) pf) (Equal n1 n2))
            freeVars f `shouldBe` Set.fromList [y]
describe "groundFormula" $ do
    it "groundFormula returns a ground formula (dependant on isGroundFormula passing
        all tests)" $ do
        property $ forAll (resize 5 arbitrary) $ \f ->
            forAll genStdNameSet $ \s ->
                all isGroundFormula (groundFormula (f :: Formula) s)
describe "checkModel" $ do
    let x = Var "x"
        px = Atom (Pred "P" [VarTerm x])
    context "checkModel satisfies validities when atoms are unground" $ do
        it "checkModel errors for P(x) -> ~~ P(x)" $ do
            property $ \m -> checkModel m (Or (Not px) (Not (Not px))) `shouldBe` True
        it "checkModel errors for x=x" $ do
```

```
        property $ \m -> checkModel m (Equal (VarTerm x) (VarTerm x)) `shouldBe`
            True
```

6 How to Use the Code

In this section we will provide instructions and examples on how to use our code.

6.1 Syntax and Semantics

To use the semantic evaluation functions from the SemanticsKL module in GHCi, begin by saving the file (e.g., SemanticsKL.lhs) in your working directory. Start GHCi by typing ghci in your terminal from that directory, then load the module with :load SemanticsKL.lhs, ensuring SyntaxKL.lhs is also present and correctly imported. After loading, you can interactively test $\mathcal{KL}$ models and formulas. For instance, create a simple world state with let ws = mkWorldState [(PPred "P" [StdName "n1"], True)] [], a model with let m = Model ws (Set.singleton ws) (Set.fromList [StdName "n1"]), and a formula like let f = Atom (Pred "P" [StdNameTerm (StdName "n1")]). Check if the formula holds using checkModel m f, which should return True since $P(n1)$ is true in the model. Alternatively, evaluate satisfiability with satisfiesModel m f.

6.2 Satisfiability and Validity Checking

To use the tableau-based satisfiability and validity checkers from the Tableau module in GHCi, first ensure the file (e.g., Tableau.lhs) is saved in your working directory. Launch GHCi from that directory by running ghci in your terminal. Load the module with :load Tableau.lhs, which will compile and make its functions available, assuming SyntaxKL.lhs and SemanticsKL.lhs are also in the same directory and properly imported. Once loaded, you can test formulas interactively. For example, define a formula like let f = Or (Atom (Pred "P" [StdNameTerm (StdName "n1")])) (Not (Atom (Pred "P" [StdNameTerm (StdName "n1")]))) and check its satisfiability with isSatisfiable f, which should return True (since $P(n1) \vee \neg P(n1)$ is satisfiable). Similarly, test validity with isValid f, which returns False. Use :reload to update changes after editing the file, and :quit to exit GHCi.

6.3 Ask and Tell

To use the AskTell module's ASK, TELL, and INITIAL operators in GHCi, set up the file as described in the previous subsections. Once loaded, you can experiment with epistemic operations. For example, define a domain with let d = Set.fromList [StdName "n1"], an initial epistemic state with let e = initial [PPred "P" [StdName "n1"]] [] [StdName "n1"], and a formula like let f = Atom (Pred "P" [StdNameTerm (StdName "n1")]). Test the ASK operator with ask d e f, which checks if $P(n1)$ is known across all worlds. Apply TELL with let e' = tell d e f to filter worlds where $P(n1)$ holds, then verify with ask d e' f, expecting True. Alternatively, use askModel and tellModel with a Model, such as let m = Model (Set.findMin e) e d, via askModel m f and tellModel m f.

6.4 Translations

You can interactively test translations and model properties. To do so, you can, for example, define a KL formula like let f = K (Atom (Pred "P" [StdNameTerm (StdName "n1")])) and translate it to standard epistemic logic (SEL) with translateFormToKr f, expecting Just (Box (P 1)). Conversely, translate an SEL formula like let g = Box (P 1) to KL with translateFormToKL g, yielding K (Atom (Pred "P" [StdNameTerm (StdName "n1")])). For models, create a KL model with let m = model1 and convert it to a Kripke model using translateModToKr m, then compare with kripkeM1 via test1. You can try out the Kripke-to-KL translation by using kripkeToKL exampleModel3 (makeWorldState 30), checking if it succeeds. You could also run some of our predefined tests like ftest1 or testAllFormulae to get to know our translation functions.

6.5 Tests

You can run all the tests and examine the current code coverage run ‘stack clean stack test –coverage‘.’